
As aulas de 1 a 5 foram elaboradas juntamente com o Prof. Ma To FU (UEM)

NOÇÕES DE CÁLCULO DIFERENCIAL E INTEGRAL

com uma variável

> # Simbolo para comentário

1 - CALCULANDO LIMITES

> # assim podemos definir uma função

> f1:=(x,y) -> (x^2-5*y)/(x^3+2*x);

> limit(f1(x,y),x=1);

Vamos ver o gráfico de uma função.

> plot((x^2-5)/(x^3+2*x),x=-2..2);



Outra forma de definir uma função é por meio do chamado procedimento. 

Um procedimento tem a forma:

P : = proc(argumentos)

local (variáveis locais)

instruções a serem executadas

end;

Vamos ver um exemplo, mais tarde voltaremos neste ponto com mais cuidado.

> salto:=proc(x); 

> if 1<=x then x^2+1 else cos(x) fi;

> end;

> salto(-1);

> salto(2);

Outro exemplo. 

> pp:=proc(x); 



> x^3;

> end;

> pp(sqrt(2));

> pp((2)^(1/3));

> # Agora vamos voltar aos limites e escrever Limit no lugar de limit. 

> #

> Limit(f1(x,2), x=1);

> value(");

Se um comando começa com letra maiúscula, então ele é "inerte". Isto é, só escreve mas não 
calcula.

> A:=sin(2*x)/x;

Observar que A não é do tipo A:=x -> sin(2*x)/x.

Portanto A não é função (é uma expressão).

> Limit(A, x=0);

> value(");

> Limit(A, x=infinity);



> value(");

Lá vão dois limites infinitos.

> Limit(1/x, x=0, left);

> value(");

Abaixo, usamos "limit" com l minúscula.

> limit(1/x, x=0, right); 

2 - CALCULANDO ALGUMAS DERIVADAS

> f2:=x -> sin(2*x);

> # Derivando uma vez.

> diff(f2(x),x);

> # Derivando 3 vezes.

> diff(f2(x),x,x,x);

> #

> # Observa que para derivar 3 vezes podemos usar x,x,x; ou $3. 

> diff(f2(x),x$3);

> #

Calculando derivadas com operador diferencial "D" A saída (output) é sempre uma função 

> h:=x -> x^2;



> h(x);

> D(h); # Vai sair uma função !

> D(h)(x);

> diff(f(x)*g(x),x); # derivada do produto

> diff( f(x,y(x)) , x ); # derivada total de f(x,y(x)) 

> #

3 - UM PROBLEMINHA DE CÁLCULO INTERESSANTE PARA DISTRAIR.

Dado um ponto P=(a,b) e o gráfico de uma função f,

calcular a distância de P ao gráfico da f.

Ao começar um novo problema costuma-se "zerar" a 

memória.

> restart:

Por definição, a distância de P ao gráfico da f é a menor distância entre os pontos Q do 
gráfico e o ponto P. Vamos então definir: 

distância=|| P-Q ||

Neste exemplo tomamos f(x)=cos(x) e P=(2,3). 

> dist:=sqrt( (2-x)^2 + (3-cos(x))^2 );

Queremos a menor distância. Da fórmula acima vemos que dist é uma função só de x. 
Portanto, do Cálculo, concluímos que a menor distância deve ser dada por 

um x que anula a derivada de dist.

Vamos ver o gráfico da função dist.



> plot( sqrt( (2-x)^2 + (3-cos(x))^2),x=0..2);

> derivada:=diff(dist,x);

Resolvendo a equação derivada=0 para "x" em [0,2]. Já vi o gráfico.

> x[min]:=fsolve(derivada, x ,0..2 );

> evalf(");

> #

Escrevendo a resposta utilizando o comando subs. (Substituir o valor de x por x_minimo na 
expressão dist).

> menor[dist]:=subs(x=x[min], dist);

> resposta:=evalf(");



> # Descubra o que foi feito (abaixo)

> L:=[ [2,3], [x[min] , cos(x[min]) ]];

> plot({cos(x), L }, x=0..3 , scaling=constrained);

4 - SÉRIES DE TAYLOR 

A função "series" escreve a 'série de Taylor de funções analíticas. Em geral, a resposta

é dada em termos de uma expansão de ordem 6.

> S:=exp(x)*x^2;

> S1:=series( S, x=0 ); # Em torno de x=0.

Queremos agora uma expansão de ordem 10.

> S2:=series( S , x=0, 10 );

Converter a série num polinômio.



> S3:=convert(S2, polynom );

> S3(2); # inútil

> value("); #inútil

Vamos transformar a expressão S3 numa função de verdade.

> P:=unapply( S3 , x );

> evalf( P(1) );

> evalf( subs(x=1, S) );

5 - CALCULANDO INTEGRAIS

> Int(ln(x),x); # Com I maiúscula

> int(ln(x),x); # Com i minúscula

> int(tan(x),x); # Com I minúscula

> Int(1/ sqrt(1-x^2),x);



> value(");

> diff(",x);

> AA:=Int(x^2*exp(x^2),x=0..1);

> resp:=value(AA);

> #

Maple utilizou a função erro "erf" que é definida por

erf(x) = 2 / sqrt(Pi) * int( exp(-t^2), t=0..x )

Mas mesmo assim podemos ter uma aproximação da integral utilizando-se evalf(AA).

> evalf(AA);

Os próximos exemplos são conhecidos.

> #

> # Integrais Múltiplas

> #

> Int( Int(x^2+y^2, y=0..2 ),x=0..2 );

> value(");



> Int( Int( Int( z*cos(x), x=0..2*y ), y=0..2 ), z=1..2 );

> value(");

> evalf(");

Veremos alguns exemplos de integrais improprias.

> f8:= exp(-u*x)*ln(x)*sqrt(x);

> int(f8,x=0..infinity);

Definite integration: Can't determine if the integral is convergent.
Need to know the sign of --> u
Will now try indefinite integration and then take limits.

Como nada é conhecido sobre u, MapleV não pode determinar 

uma resposta. Podemos usar o comando ASSUME para informar 

ao MapleV sobre u.

> assume(u<0); int(f8,x=0..infinity);

Esta integral diverge. Por outro lado, se u>0 temos convergência e uma resposta simbólica.

> assume(u>0);int(f8,x=0..infinity);

> #



> Int(1/x^2 , x=1..infinity);

> value(");

> #

> # Outro exemplo.

> #

> int(1/x, x=0..1);

> #

> # Observe isto !

> #

> int(1/(x^3), x=a..b);

> subs(a=-1, b=4, " );

> # Ou ainda:

> int(1/(x^3), x=-1..4, 'CauchyPrincipalValue' );

Um exemplo sobre transformada de Laplace.

> restart; with(inttrans):

> laplace(t**3-cos(t)=y(t), t, s); 



> laplace(t^(1/2)-exp(-t)+sinh(a*t), t, s);

> 


